

SAS® Statistical Graphics Procedures

Sanjay Matange, Dan Heath, SAS Institute Inc.
SAS Global Forum 2013

An effective graph can reveal trends or patterns in your data that may have otherwise remained hidden in a tabular form. SAS 9.2 introduces a new family of procedures for creation of statistical graphs ranging from simple scatter plots to paneled displays to assist in data analysis.

This family of SG procedures includes the SGPLOT, SGSCATTER and SGPANEL procedures. The SG procedures provide a concise syntax to create graphs commonly used in data analysis in many industries.


The SG procedures use the Graph Template Language (GTL) behind the scenes to create the graphs. If you need complex graphs that


are beyond the scope of these procedures, you may use GTL directly. For more information on GTL, please see the handout: "SAS/GRAPH: Graph Template Language".

With SAS 9.3, new features have been added for Cluster Groups, Box Plots on interval axes, Bubble Plots, High Low plots, Attribute Maps and Annotation shown on page 3.

With SAS 9.4, new features are added for assigning group colors, splitting axis values, jittering, proportional column widths, etc. The AXISTABLE statement makes it easy to include axis aligned textual data in a graph.

These examples and their complete programs can be viewed under the link for each procedure at: <http://support.sas.com/rnd/datavisualization/yourGraphs/analyticalCustom/sqf2013/>






```

title "Fitness Profile (Pulse)";
proc sgscatter data=fitsort;
  matrix runpulse rstpulse maxpulse age /
    diagonal=(histogram normal) group=group;
run;


```



```

title "Leading Causes of US Cancer Deaths in 2007";
proc sgplot data=cancer;
  hbar cause / response=mcases fillattrs=graphdata1
    transparency=.65 legendlabel="New Cases (Male)"
    name="mcases" ;
  hbar cause / response=mdeaths barwidth=.5
    fillattrs=graphdata1 transparency=.25
    legendlabel="Deaths (Male)" name="mdeaths" ;
  hbar cause / response=fcases fillattrs=graphdata2
    transparency=.65 legendlabel="New Cases (Female)"
    name="fcases" ;
  hbar cause / response=fdeaths barwidth=.5 fillattrs=graphdata2
    transparency=.25 legendlabel="Deaths (Female)"
    name="fdeaths" ;
  yaxis display=(nolabel) discreteorder=data; run;


```



```

title "Auto Performance Statistics";
proc sgscatter data=sashelp.cars;
  where type='Sedan';
  plot (mpg_city weight) * (enginesize horsepower) /
    group=origin reg=(nogroup clm degree=2) grid ;
run;

```




```

title 'Emerging Issues';
proc sgplot data=wu2 noautolegend nocycleattr;
  scatter x=x y=y / group=z1 markerattrs=(symbol=squarefilled
    size=8) name='s1' nomissinggroup;
  scatter x=x y=y / group=z markerattrs=(symbol=squarefilled
    size=20) name='s3' nomissinggroup;
  scatter x=x y=y / markerattrs=(symbol=square size=20);
  xaxis type=time label='Build Periods'
    offsetmin=0.1 offsetmax=0.1;
  yaxis type=discrete label='Periods in Service';
  keylegend 's1' / location=outside position=right across=1;
run;


```

SAS® 9.3 Features




```

title 'Adverse Event Timeline for Patient Id xxxxxxxx';
proc sgplot data=AETimelinecap dattmap=attmap;
  highlow y=aeseq low=aestdate high=aeendate /
    type=bar group=aesev barwidth=0.8
    lineattrs=(color=black thickness=2) lowlabel=aeedecod
    highcap=highcap attrid=Severity labelattrs=(size=8) ;
  xaxis grid display=(nolabel);
  yaxis grid display=(noticks novalues nolabel);
run;
  
```




```

title "Mileage by Type and Origin";
footnote j=1 "Excluding Hybrids";
proc sgplot data=sashelp.cars;
  where type ne "Hybrid";
  vbar type / response=mpg_city group=origin stat=mean
    groupdisplay=cluster nostatlabel dataskin=sheen;
  xaxis display=(nolabel);
  yaxis grid;
run;
  
```



```


proc sgplot data=QTcData sganno=anno
  pad=(bottom=12%);
  format week qtcweek.;
  vbox qtc / category=week group=drug
    groupdisplay=cluster meanattrs=(color=black);
  refline 26 / axis=x;
  refline 0 30 60 / axis=y lineattrs=(pattern=shortdash);
  xaxis type=linear values=(0 2 4 8 12 16 20 24 28)
    max=29 valueshint;
  yaxis label='QTc change from baseline'
    values=(-120 to 90 by 30);
  keylegend / title="";
run;
  
```




```

title 'Student Age by Height and Weight';
proc sgplot data=sashelp.class;
  bubble x=height y=weight size=age / group=sex
    transparency=0.2;
  inset 'Bubble size is proportional to age' /
    position=bottomright;
  xaxis grid; yaxis grid;
  keylegend / location=inside position=topleft;
run;
  
```

SAS® 9.4 Features



This graph uses proportional, sort and axis value split options




This graph uses the external axis table to display the at-risk values

```


title 'Mean MSRP for Luxury Sedans by Make and Origin';
proc sgpanel data=LuxurySedans;
where type = 'Sedan' and msrp > 40000;
panelby origin / proportional uniscale=row novarname
    layout=columnlattice onepanel sort=ascmean;
vbar make / response=msrp dataskin=gloss stat=mean
    group=origin datalabel
    categoryorder=respsc transparency=0.2;
colaxis display=(nolabel) valueattrs=(size=7);
rowaxis valueattrs=(size=7);
run;
    
```

```

proc sgplot data=sasuser.survivalplotdata;
step x=time y=survival / group=stratumnum
    lineattrs=(thickness=2) curvelabel;
scatter x=time y=censored /
    markerattrs=graphdatadefault(symbol=plus) name='c';
scatter x=time y=censored / group=stratum
    markerattrs=(symbol=plus);
xaxistable atrisk / x=tatrisk class=stratumnum location=inside
    colorgroup=stratum valueattrs=(size=9);
keylegend 'c' / location=inside position=topright;
keylegend 's' / autoitemsize across=3; run;
    
```



This graphs uses jitter on discrete X axis



This graph uses the internal axis table statement. Data: Mother Jones

```

title 'Cholesterol by Death Cause';
footnote 'This graphs uses JITTER on discrete X axis';
proc sgplot data=heart;
scatter x=deathcause y=cholesterol / jitter
    markerattrs=graphdata1(size=9 symbol=circlefilled)
    filledoutlinedmarkers markeroutlineattrs=(thickness=0)
    markerfillattrs=graphdata1 dataskin=sheen;
xaxis display=(nolabel) valueattrs=(size=7) offsetmin=0.15;
yaxis grid;
run;
    
```

```

proc sgplot data=USMassKillingSorted(obs=20) noautolegend;
styleattrs datacontrastcolors=(black red);
hbarparm category=id response =total_victims /
    datalabel = total_victims transparency=0.2
    datalabelattrs=(size=5 color=purple) fillattrs=graphdata3;
series y=id x=total_victims / lineattrs=(thickness=0) x2axis;
yaxistable year case signs death_injury / label location=inside
    position=left valueattrs=(size=6) labelattrs=(size=7)
    colorgroup=weapons_legal;
yaxistable venue weapons_legal race / label location=inside
    valueattrs=(size=6) labelattrs=(size=7)
    position=right colorgroup=weapons_legal;
xaxis display=(novalues noticks nolabel);
x2axis display=(noticks novalues) labelattrs=(weight=bold);
yaxis display=none colorbands=even discreteorder=data
    colorbandsattrs=(transparency=0.9 color=darkred); run;
    
```